/*
 * Licensed under the MIT License
 * 
 * Copyright (c) 2010 Kay Kreuning
 * 
 * Permission is hereby granted, free of charge, to any person obtaining a copy
 * of this software and associated documentation files (the "Software"), to deal
 * in the Software without restriction, including without limitation the rights
 * to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the Software is
 * furnished to do so, subject to the following conditions:
 * 
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 * 
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
 * THE SOFTWARE.
 * 
 * http://code.google.com/p/as3hyvesapi
 */
package nl.hyves.api.utils
{
	import nl.hyves.api.sessions.OAuthToken;

	import flash.net.SharedObject;

	public class TokenStore
	{
		public static function clear(id:String):Boolean
		{
			SharedObject.getLocal(id).clear();

			return !hasValid(id);
		}

		public static function hasValid(id:String, methods:Array = null):Boolean
		{
			var data:Object = SharedObject.getLocal(id).data;
			
			if (data && data.type == OAuthToken.ACCESSTOKEN && data.expireDate)
			{
				if (methods && AuthUtil.calculateSessionSignature(methods) != data.signature)
				{
					return false;
				}
				
				var expireTimestamp:Number = data.expireDate.getTime();
				var currentTimestamp:Number = new Date().getTime() - 1000;	// Subtract one second for networking delay.
				
				return (expireTimestamp > currentTimestamp);
			}

			return false;
		}

		public static function get(id:String):OAuthToken
		{
			if (hasValid(id))
			{
				var data:Object = SharedObject.getLocal(id).data;

				return new OAuthToken(data.type, data.token, data.secret, data.userId, data.methods, data.expireDate);
			}

			return null;
		}

		public static function set(id:String, token:OAuthToken):Boolean
		{
			var data:Object = SharedObject.getLocal(id).data;
			data.type = token.type;
			data.token = token.token;
			data.secret = token.secret;
			data.userId = token.userId;
			data.methods = token.methods;
			data.expireDate = token.expireDate;
			data.signature = AuthUtil.calculateSessionSignature(token.methods);

			return hasValid(id, token.methods);
		}
	}
}